// 编译：240884432
// 简称: vip09
// 名称: vip09
// 类别: 公式应用
// 类型: 用户应用
// 输出: Void
//------------------------------------------------------------------------
Params
	//默认螺纹8881小时
	Numeric M(20); //多均线参数 参数范围及步长： 5-100，5
	Numeric N(20);	//轨道区间参数 参数范围及步长：5-100，5
	Numeric X(7); //OBV指标平滑系数 参数范围及步长：1-20，1
	Numeric TRS(75);//移动止损止盈幅度 参数范围及步长： 5-100，5
	Numeric Fund(20000); //投入保证金；
Vars
	Numeric Lots(0);
	Series<Numeric> OBVValue;
	Series<Numeric> MA1;
	Series<Numeric> MA2;
	Series<Numeric> MA3;
	Series<Numeric> MAOBV;
	Series<Numeric> M2;
	Series<Numeric> cond1;
	Series<Numeric> cond2;
	Series<Numeric> cond3;
	Series<Numeric> cond4;
	Series<Numeric> cond5;
	
	Series<Numeric> kong_cond1;
	Series<Numeric> kong_cond2;
	Series<Numeric> kong_cond3;
	Series<Numeric> kong_cond4;
	Series<Numeric> kong_cond5;
	
	Series<Bool> Long_line;
	Series<Bool> short_line;
	Series<Numeric> bandup;
	Series<Numeric> banddn;
	Series<Numeric> MyPrice;//开仓价格
	Series<Numeric> MyPrice2;//开仓价格
	Series<Numeric> HighAfterEntry;//开仓后出现的最高价
	Series<Numeric> LowAfterEntry;//开仓后出现的最低价
	Series<Numeric> liQKA;
	Series<Numeric> DliqPoint;
	Series<Numeric> KliqPoint;
	Series<Numeric> barcoutN;
	Series<Numeric> bar_entry_count;
	Series<Numeric> kaicang_kg;
		//Bar更新事件函数，参数indexs表示变化的数据源图层ID248844320数组

Events
	//此处实现事件函数
	
	//初始化事件函数，策略运行期间，首先运行且只有一次，应用在订阅数据等操作
	OnInit()
	{
		//与数据源有关
		Range[0:DataCount-1]
		{
			//=========数据源相关设置==============
			AddDataFlag(Enum_Data_RolloverBackWard());	//设置后复权

			AddDataFlag(Enum_Data_RolloverRealPrice());	//设置映射真实价格

			AddDataFlag(Enum_Data_AutoSwapPosition());	//设置自动换仓

			//AddDataFlag(Enum_Data_IgnoreSwapSignalCalc());	//设置忽略换仓信号计算
		}

	}

	//Bar更新事件函数，参数indexs表示变化的数据源图层ID数组
	OnBar(ArrayRef<Integer> indexs)
	{
		// 记录开仓后的最高点和最低点
		If(bar_entry_count == 0)
		{
			HighAfterEntry = High;  // 如果是第一次开仓，则记录开仓时的最高价
			LowAfterEntry = Low;    // 如果是第一次开仓，则记录开仓时的最低价
		}else
		{
			HighAfterEntry = Min(HighAfterEntry,High); // 空头止损，更新最低的最高价
			LowAfterEntry = Max(LowAfterEntry,Low);    // 多头止损，更新最高的最低价
		}
		Lots=Max(1,IntPart(Fund/(O*ContractUnit*BigPointValue*0.1)));    // 计算开仓手数
		If(CurrentBar == 0)
		{
			OBVValue = 0;   // 如果是第一个Bar，则OBV值初始化为0
		}Else
		{
			If(Close>Close[1])
			{
				OBVValue = OBVValue[1]+Vol;    // 如果收盘价比前一根Bar高，则OBV增加当日成交量
			}Else If(Close<Close[1])
			{
				OBVValue = OBVValue[1]-Vol;    // 如果收盘价比前一根Bar低，则OBV减少当日成交量
			}Else
			{
				OBVValue = OBVValue[1];    // 如果收盘价和前一根Bar相等，则OBV保持不变
			}
		}
		MA1=Average(C,M);   // 计算C的M周期移动平均线
		MA2=Average(C,M*2); // 计算C的2*M周期移动平均线
		MA3=Average(C,M*3); // 计算C的3*M周期移动平均线
		MAOBV=Average(OBVValue,M);  // 计算OBV值的M周期移动平均线
		M2=Average(OBVValue,X); // 计算OBV值的X周期移动平均线
		cond1=IIF(MA1>MA1[1] OR C>MA1,MA1,cond1);  // 根据条件判断更新cond1的值
		cond2=IIF(M2>M2[1],MA1,cond2);   // 根据条件判断更新cond2的值
		cond3=IIF(C>MA2 and MAOBV>MAOBV[1],MA2,cond3);   // 根据条件判断更新cond3的值
		cond4=IIF(C>MA2,MA2,cond4); // 根据条件判断更新cond4的值
		cond5=IIF(C>MA3 and MAOBV>MAOBV[1],MA3,cond5);   // 根据条件判断更新cond5的值

		kong_cond1=IIF(MA1<MA1[1] OR C<MA1,MA1,kong_cond1);   // 根据条件判断更新kong_cond1的值
		kong_cond2=IIF(M2<M2[1],MA1,kong_cond2); // 根据条件判断更新kong_cond2的值
		kong_cond3=IIF(C<MA2 and MAOBV<MAOBV[1],MA2,kong_cond3);   // 根据条件判断更新kong_cond3的值
		kong_cond4=IIF(C<MA2,MA2,kong_cond4);   // 根据条件判断更新kong_cond4的值
		kong_cond5=IIF(C<MA3 and MAOBV<MAOBV[1],MA3,kong_cond5);   // 根据条件判断更新kong_cond5的值

		PlotNumeric("cond1",cond1);    // 绘制cond1的数值
		PlotNumeric("cond2",cond2);    // 绘制cond2的数值
		PlotNumeric("cond3",cond3);    // 绘制cond3的数值
		PlotNumeric("cond4",cond4);    // 绘制cond4的数值
		PlotNumeric("cond5",cond5);    // 绘制cond5的数值
		bandup=Highest(H,N);    // 计算N周期内的最高价
		banddn=Lowest(L,N); // 计算N周期内的最低价
		Long_line=C>MA3 and MAOBV>MAOBV[1];    // 判断是否满足多头条件
		short_line=C<MA3 and MAOBV<MAOBV[1];   // 判断是否满足空头条件

		if(Long_line[1])
		{
			if(cond4[1]>cond5[1] and cond5[1]>cond5[2] and cond4[1]>cond4[2] and cond3[1]>cond3[2] and cond2[1]>cond2[2] and cond1[1]>cond1[2]) //
			{
				if(MarketPosition!=1 and H>=bandup[1])
				{
					Buy(Lots,max(open,bandup[1])); // 如果满足开多条件，则买入
					LowAfterEntry = EntryPrice;//保存多头开仓价格;
					kaicang_kg=1;   // 标记已开仓
					bar_entry_count=0;  // 重置连续未开仓的Bar计数
				}
			}
		}
		if(short_line[1])
		{
			if(kong_cond4[1]<kong_cond5[1] and kong_cond5[1]<kong_cond5[2] and kong_cond4[1]<kong_cond4[2] and kong_cond3[1]<kong_cond3[2] and kong_cond2[1]<kong_cond2[2] and kong_cond1[1]<kong_cond1[2]) //
			{
				if(MarketPosition!=-1 and L<=banddn[1])
				{
					SellShort(Lots,min(open,banddn[1])); // 如果满足开空条件，则卖空
					HighAfterEntry = EntryPrice;//保存空头开仓价格;
					kaicang_kg=1;   // 标记已开仓
					bar_entry_count=0;  // 重置连续未开仓的Bar计数
				}
			}
		}
		
		if(kaicang_kg[1]==1) // 如果在本周期内有开仓操作
		{
			bar_entry_count=bar_entry_count+1; // 将开仓计数器加一
		}
		Else
		{
			bar_entry_count=0; // 如果没有开仓，则将开仓计数器重置为 0
		}
		
    	//移动出场
    	If(MarketPosition == 0)   // 自适应参数默认值；
    	{
    		liQKA = 1; 
    		barcoutN=0;
    	}Else if(bar_entry_count>barcoutN)					 //当有持仓的情况下，liQKA会随着持仓时间的增加而逐渐减小，即止损止盈幅度乘数的减少。
    	{
    		liQKA = liQKA - 0.1; 
    		liQKA = Max(liQKA,0.3);
    		barcoutN=bar_entry_count;
    	}
    	if(MarketPosition>0)
    	{
			DliqPoint = LowAfterEntry - (Open*TRS/1000)*liQKA; //经过计算，这根吊灯出场线会随着持仓时间的增加变的越来越敏感；
    	}
    	if(MarketPosition<0)
    	{
			KliqPoint = HighAfterEntry + (Open*TRS/1000)*liQKA; //经过计算，这根吊灯出场线会随着持仓时间的增加变的越来越敏感；
    	}
    
    	If(KliqPoint[1]>0 and MarketPosition<0)PlotNumeric("KliqPoint[1]",KliqPoint[1]);
    	if(DliqPoint[1]>0 and MarketPosition>0)PlotNumeric("DliqPoint[1]",DliqPoint[1]);
    	
    	// 持有多单时
    	If(MarketPosition >0 And bar_entry_count >0  And Low <= DliqPoint[1] and DliqPoint[1]>0 and DliqPoint[1]>0  and bar_entry_count>0) 
    	{
    	
    			Sell(0,Min(Open,DliqPoint[1]));
    			DliqPoint=0;
    			barcoutN=0; 
    			kaicang_kg=0;
    			bar_entry_count=0;
    			
    			
    	}
    	// 持有空单时
    	If(MarketPosition <0 And bar_entry_count >0  And High >= KliqPoint[1] and KliqPoint[1]>0 and KliqPoint[1]>0  and bar_entry_count>0)
    	{	
    		
    			BuyToCover(0,Max(Open,KliqPoint[1]));
    			KliqPoint=0;
    			barcoutN=0;
    			kaicang_kg=0;
    			bar_entry_count=0;

    	}
	
	}

